
Klaus Kohl-Schöpe 

Chess Board auf dem WikiReader 

Da ich mich für Schach interessiere und einige Bücher mit in den Urlaub nehmen 
wollte, suchte ich nach einer Möglichkeit, die Züge aus den Büchern 
nachzuvollziehen. Aber da ich dann meist den ganzen Tag in der Sonne wäre, 
war das Handy keine Option, sondern es sollte ein Low-Power LCD Gerät sein. 
Glücklicherweise ist mir eingefallen, dass ich noch ein paar WikiReader habe, die 
genau dem Wunsch entsprechen und auch in Forth programmierbar sind. 

Der WikiReader 

ist ein Projekt aus dem Jahr 2009, um 
Inhalte des Wikipedia offline zu lesen 
(siehe Bild 1). Es basiert auf den Epson 
S1C33E07 Microcontroller mit 64KB Flash, 
32MB SDRAM und einer microSD Karte. 
Das Gerät ist mit 10x10x2cm³ klein, hat ein 
240x208 Pixel Touch-Display und drei 
Tasten. Zwei AAA-Batterien sollten für 90h 
Betrieb (1300h lesen) ausreichen. Neben 
Anzeige des Wikipedia oder Bücher des 
Gutenberg-Projektes kann es auch Forth-
Programme ausführen. 

 
Bild 1 – WikiReader 

Leider wurde 2014 das Projekt ohne 
Begründung eingestellt, aber ich konnte 
mir Dank Dirk Brühl noch einige Exemplare 
bei amazon.com sichern. Auch wird es 
immer schwieriger, neben den Sourcecode 
auf Github weitere Unterlagen dazu zu 
finden. Glücklicherweise gibt einen Forth-
Simulator für Windows, welche ich für die 
Entwicklung nutzte. 

Das Forth 

Christopher Hall von Openmoko hat ein 
fast ANSI-konformes Forth mit einem 8x13-
Zeichensatz implementiert, welches beim 
beim Einschalten mit gedrückter random-
Taste ein Menü mit Auswahl von Pro-
grammen aus dem File bootmenu.4th auf 
der SD-Karte anzeigt. 

 
Bild 2 – Forth-Menü 

Neben diversen Testprogrammen und 
einem Rechner ist hier ist auch schon mein 
Programm CHESS gelistet, welches ich 
vorstellen will. Es sollte vorerst nicht 
vollständig Schach spielen können, 
sondern nur Züge anzeigen oder wieder 
zurücknehmen können und sieht in der 
Anwendung aktuell so aus: 

 
Bild 3 – Schachbrett 

Damit kommen wir auch schon zu den 
verschiedenen Aufgaben, die mit einem 
solchen Programm verbunden sind:  
- Erstellung des Zeichensatzes  
- Zugeingabe über Touch  
- Zugrücknahme 

Der Zeicheneditor 

Da der Bildschirm 208 Pixel hoch ist und 
auch noch die Tastenbelegung und die 
Züge angezeigt werden sollten, nutze ich 
24x24 Pixel pro Feld. Mit 8 unter-
schiedlichen Figuren in zwei Farben auf 
weißen und schwarzen Felder braucht man 
32 (+2 für leere Felder) Bilder. Ich bin nicht 
besonders stolz, was ich hier gezeichnet 
habe, aber das Programm war in einer 
Woche realisiert und die Figuren unter-
scheidbar. 

Da ich damals (während KKForth-Zeiten ab 
1990) bei meinen Screeneditor auch unter-
schiedliche Zeilenanzahl und -länge 
eingeben konnte, habe ich das C64-Format 
40x25 für den Zeicheneditor gewählt. 
Damit bleiben neben den 24 Pixel auch 
Platz für den Hex-Code (Bild 4). Ich muss 
nur die 24 Zeichen pro Zeile lesen und in 3 
Bytes wieder speichern. 

 
Bild 4: IconEditor 

Die Routine zur Berechnung des Hex-
Codes (hexicons) und deren Speicherung 
in das Programmfile (saveicons <name>) 
sind auch in diesem ChessIcon.scr-File 
aber mit üblicher 16x64-Formatierung 
(siehe Listing 1). Das Ergebnis habe ich 
dann als Array in mein Chess.4th-
Programm übernommen. 

Der Simulator 

Schon meine Arbeiten mit FPGAs zeigte 
mir, dass eine Entwicklung auf der 
Zielhardware schwierig und oft langsam ist. 
Vor allem wenn man Programme über 
serielle Schnittstelle oder sogar als Image 
auf eine SD-Karte brennen muss. Deshalb 
war ich froh, dass es für dieses 
WikiReader-Forth einen Windows-
Simulator gibt. Es gibt zwar kleine 
Unterschiede (z.B. das fehlende cmove), 

die man aber leicht korrigieren kann. 

Im Endeffekt muss man nur drei Files im 
Simulatorverzeichnis bzw. auf der SD-
Karte editieren oder hinzufügen. Als erstes 
muss man File „bootmenu.4th“ um  
    s" CHESS.4MU"    add-item  
im Befehl read-items ergänzt. Das hier 

angegebene CHESS.4MU ist sehr kurz: 

include chess.4th 

chess 

Es ladet dann das eigentliche Programm 
CHESS.4TH, was im WikiReader ein paar 
Sekunden dauern kann. 

Test im WikiReader 

Im WikiReader kann man natürlich auch 
interaktiv mit dem Forth arbeiten, aber 
dazu benötigt man eine serielle Schnitt-
stelle, die man mit etwas Geschick an dem 
Batteriefach herausführen kann (Siehe 
Link 4). Man muss sowieso das Gehäuse 
öffnen, um an die SD-Karte zu kommen 
und dort die drei oben angegebenen Files 
zu modifizieren bzw. dort abzulegen. 

Das Programm 

Die oben erstellten Icons sind der erste Teil 
des Programmes, wobei immer die 24*24 
Pixel in 9 Bytes mal 8 Zeilen (72 Bytes) 
zusammengefasst werden. 

Natürlich wollte ich auch die Züge 
speichern, um sie bei Bedarf auch zurück-
zunehmen. Deshalb merke ich mir bei 

jedem Zug die Ausgangsposition, die ge-
zogene Figur, die Endposition und die 
ersetze Figur. 

Als nächstes gibt es Arrays mit der 
Startaufstellung und das Array mit der 
aktuellen Aufstellung, welche die Routine 
drawboard auf das LCD zeichnet. Als 

Ergänzung werden auch noch die letzten 
10 Züge rechts neben dem Board 
aufgelistet. 

Beim Zeichnen ist zu beachten, dass zwar 
auch ein Bit pro Pixel verwendet wird, aber 
bei einer Zeile mit 240 Pixel (30 Byte) zwei 
zusätzliche Bytes reserviert sind, welche 
nicht angezeigt werden. Natürlich muss 
man auch aufpassen, was bei der 
Nummerierung des Feldes auf dem 
Schachbrett und dann im Display oben 
bzw. unten ist. Teilweise hatte ich 
unerklärliche Effekte, weil Pixel nicht richtig 
dargestellt oder überschrieben wurden. 

Bei den Experimenten mit Touch zeigte 
sich, dass eine Abfrage sinnvoll ist, ob das 
gewählte Feld auch mit der Figur der 
richtigen Farbe besetzt ist. Es wird dann 
das Feld markiert und darauf gewartet, bis 
das Ziel gewählt wird. Danach wird geprüft, 
ob das Zielfeld nicht mit der eigenen Farbe 
belegt ist, was ja nicht erlaubt ist. Ist das ok 
(weitere mögliche Abfragen im nächsten 
Kapitel) wird der Zug gespeichert und die 
Anzeige aktualisiert. 

Damit fehlt nur noch der Hauptbefehl 
chess . Dieses übernimmt die Startauf-

stellung in die aktuelle Konfiguration, setzt 
den Zügezähler zurück und wartet auf die 
Eingabe. Diese können die Tasten für 
Zurück bzw. Vorwärts sein oder das 
Programm beenden. Über das Touch-
Display werden dann die Züge einge-
geben, wobei auch da zuerst geprüft wird, 
ob ein Schachfeld angeklickt wurde. Bei 
der ersten Berührung wird geprüft, ob die 
erlaubte Figur gewählt wurde und bei der 
zweiten Berührung dann (leider noch 
unvollständig) die erlaubten Ziele. 

Notwendige Verbesserungen 

Leider habe ich dann das Projekt nicht 
mehr weitergemacht, obwohl noch viele 
Punkte notwendig wären:  
- Vollständige Prüfung, ob Zug erlaubt ist 
- Rochade und En passant  
- Bauernumwandlung  
- Editor für Aufstellung 

Auch ein komplettes Schachprogramm mit 
Eröffnungsbibliothek wäre hier möglich, 
aber dies werde ich vermutlich mit einer 
deutlich schnelleren Hardware realisieren. 

Ich denke, dieses Beispiel zeigt, dass man 
manchmal auch alte Hardware für eine 
kleine Applikation heranziehen kann, wenn 
man spezielle Anforderungen hat.  

Viel Spaß damit wünscht 
Klaus Kohl-Schöpe 
 

 
 
Links: 

1. WikiReader: 
https://de.wikipedia.org/wiki/WikiReader 

2. Offizielle WikiReader Sourcen im Github:  
https://github.com/wikireader/wikireader 

3. WikiReader FORTH Simulator:  
http://createuniverses.blogspot.com/2011/03/wikireader-forth-simulator.html 

4. WikiReader Serial Port Hack:  
http://toddbot.blogspot.com/2010/05/wikireader-serial-port-hack-pictures.html 

 
  

https://de.wikipedia.org/wiki/WikiReader
https://github.com/wikireader/wikireader
http://createuniverses.blogspot.com/2011/03/wikireader-forth-simulator.html
http://toddbot.blogspot.com/2010/05/wikireader-serial-port-hack-pictures.html

Listing 1: Sourcen für IconEditor: 
Sourcen sind für mcForth. 

Besonderheit: Create und Createp muss mit End-Create abgeschlossen werden 

 

Screen 1: 

\ Loadscreen with Converter 

  : readblock    ( n -- addr ) \ read block to here 

   #1024 um* 0 fid @ fpos! here dup >x #1024 fid @ fread drop ; 

  : writeblock    ( n -- ) \ write block from here 

   #1024 um* 0 fid @ fpos! here     >x #1024 fid @ fwrite ; 

  : hexicons    ( -- ) \ Convert icons to hex code 

    base push  hex 

    capacity 1 -  3 ?DO                 \ for all blocks 

      i readblock  #40 +  #24 FOR       ( -- addr ) \ all lines 

        dup #26 +  3 FOR                ( -- addr addr+26 ) 

          0 swap  8 FOR                 ( -- addr 0 addr+26 ) 

            >r >r  dup c@ #32 - IF #219 over c!  1 ELSE 0 THEN 

               r> 2* +  swap 1+ swap r> ( -- addr+1 n addr+26 ) 

          NEXT >r 0 <# # # [char] $ hold #> r@ swap cmove r> 4 + 

        NEXT drop #16 +                 ( -- addr+24+16 ) 

    NEXT drop  i writeblock  LOOP ; 

 

Screen 2: 

\ Save Icons 

  Createp crlf  $0d    c,p  $0a    c,p          End-Create 

  Createp "c,"  char c c,p  char , c,p  $20 c,p End-Create 

  : saveicons   ( file ; -- ) 

    parse-name  $>h count  fcreate           \ new file 

    capacity 1 -  3 ?DO                      \ for all blocks 

  i readblock over >r  dup #40 -trailing >r >x r> r> fwrite 

      #66 +   over >r  crlf p>x 2 r> fwrite 

      #8 FOR 

        3 FOR  3 FOR 

              over >r  dup    >x 4 r> fwrite  4 + 

              over >r  "c,"  p>x 3 r> fwrite 

          NEXT  #28 + 

        NEXT  over >r  crlf  p>x 2 r> fwrite 

      NEXT drop 

    LOOP fclose ;                   \ close file 

 
Listing 2: Sourcen für Programm Chess im WikiReader: 
\ ------------------------------------------------------------- 

\ Chess for Wikireader 

\ Author: Klaus Kohl-Schoepe 

\ ------------------------------------------------------------- 

\ Description: 

\  Display: Start at lcd-vram with 32 byte width (30 visible) 

\  Board: 192x192 Pixel (24*24 fields) with text on the right 

\ ------------------------------------------------------------- 

 

base @  decimal 

 

\ Missing on Simulator 

: cmove 

  0 ?do over c@  over c!  1+ swap 1+ swap loop drop drop ; 

 

\ ------------------------------------------------------------- 

\ Chess Icons 

\ ------------------------------------------------------------- 

\ Figures: *2 for white field and *2+1 for black field 

\  0=empty; 1/2=white/black pawn; 3/4=bishop 

\  4/5=knight; 6/7=rook, 8/9=bishop, 10/11=king 

 

24 constant fieldsize 

 

create icons 

\ 0a: White field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 0b: Black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $55 c, $55 c, $AA c, $AA c, $AA c, $55 c, $55 c, $55 c,  

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $55 c, $55 c, $AA c, $AA c, $AA c, $55 c, $55 c, $55 c,  

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $55 c, $55 c, $AA c, $AA c, $AA c, $55 c, $55 c, $55 c,  

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $55 c, $55 c, $AA c, $AA c, $AA c, $55 c, $55 c, $55 c,  

\ 1a: White pawn on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $18 c, $00 c, $00 c, $3C c, $00 c, $00 c, $66 c, $00 c,  

$00 c, $66 c, $00 c, $00 c, $3C c, $00 c, $00 c, $FF c, $00 c,  

$01 c, $81 c, $80 c, $00 c, $FF c, $00 c, $00 c, $66 c, $00 c,  

$00 c, $66 c, $00 c, $00 c, $FF c, $00 c, $01 c, $81 c, $80 c,  

$03 c, $FF c, $C0 c, $03 c, $00 c, $C0 c, $03 c, $FF c, $C0 c,  

$01 c, $FF c, $80 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 1b: White pawn on black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $55 c, $55 c, $AA c, $AA c, $AA c, $55 c, $41 c, $55 c,  

$AA c, $98 c, $AA c, $55 c, $3C c, $55 c, $AA c, $66 c, $2A c,  

$54 c, $66 c, $55 c, $AA c, $3C c, $2A c, $54 c, $FF c, $15 c,  

$A9 c, $81 c, $AA c, $54 c, $FF c, $15 c, $AA c, $66 c, $2A c,  

$54 c, $66 c, $15 c, $A9 c, $FF c, $AA c, $51 c, $81 c, $95 c,  

$AB c, $FF c, $CA c, $53 c, $00 c, $D5 c, $AB c, $FF c, $CA c,  

$51 c, $FF c, $95 c, $A8 c, $00 c, $2A c, $55 c, $55 c, $55 c,  

\ 2a: Black pawn on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $18 c, $00 c, $00 c, $3C c, $00 c, $00 c, $6E c, $00 c,  

$00 c, $7E c, $00 c, $00 c, $3C c, $00 c, $00 c, $FF c, $00 c,  

$01 c, $BD c, $80 c, $00 c, $FF c, $00 c, $00 c, $6E c, $00 c,  

$00 c, $6E c, $00 c, $00 c, $FF c, $00 c, $01 c, $99 c, $80 c,  

$03 c, $FF c, $C0 c, $03 c, $3C c, $C0 c, $03 c, $FF c, $C0 c,  

$01 c, $FF c, $80 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 2b: White pawn on black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $55 c, $55 c, $AA c, $AA c, $AA c, $55 c, $41 c, $55 c,  

$AA c, $98 c, $AA c, $55 c, $3C c, $55 c, $AA c, $6E c, $2A c,  

$54 c, $7E c, $55 c, $AA c, $3C c, $2A c, $54 c, $FF c, $15 c,  

$A9 c, $BD c, $AA c, $54 c, $FF c, $15 c, $AA c, $6E c, $2A c,  

$54 c, $6E c, $15 c, $A9 c, $FF c, $AA c, $51 c, $99 c, $95 c,  

$AB c, $FF c, $CA c, $53 c, $3C c, $D5 c, $AB c, $FF c, $CA c,  

$51 c, $FF c, $95 c, $A8 c, $00 c, $2A c, $55 c, $55 c, $55 c,  

\ 3a: White bishop on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $18 c, $00 c,  

$00 c, $3C c, $00 c, $00 c, $66 c, $00 c, $00 c, $DB c, $00 c,  

$01 c, $BD c, $80 c, $01 c, $BD c, $80 c, $00 c, $DB c, $00 c,  

$00 c, $66 c, $00 c, $00 c, $3C c, $00 c, $00 c, $7E c, $00 c,  

$00 c, $66 c, $00 c, $00 c, $C3 c, $00 c, $00 c, $C3 c, $00 c,  

$01 c, $81 c, $80 c, $01 c, $81 c, $80 c, $03 c, $00 c, $C0 c,  

$07 c, $FF c, $E0 c, $07 c, $00 c, $E0 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 3b: White bishop on black field 

$AA c, $AA c, $AA c, $55 c, $41 c, $55 c, $AA c, $98 c, $AA c,  

$55 c, $3C c, $55 c, $AA c, $66 c, $2A c, $54 c, $DB c, $15 c,  

$A9 c, $BD c, $AA c, $55 c, $BD c, $95 c, $A8 c, $DB c, $2A c,  

$54 c, $66 c, $55 c, $AA c, $3C c, $AA c, $55 c, $7E c, $55 c,  

$AA c, $66 c, $2A c, $54 c, $C3 c, $55 c, $A8 c, $C3 c, $2A c,  

$55 c, $81 c, $95 c, $A9 c, $81 c, $8A c, $53 c, $00 c, $C5 c,  

$A7 c, $FF c, $EA c, $57 c, $00 c, $E5 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $2A c, $55 c, $55 c, $55 c,  

\ 4a: Black bishop on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $18 c, $00 c,  

$00 c, $3C c, $00 c, $00 c, $7E c, $00 c, $00 c, $E7 c, $00 c,  

$01 c, $C3 c, $80 c, $01 c, $C3 c, $80 c, $00 c, $E7 c, $00 c,  

$00 c, $7E c, $00 c, $00 c, $3C c, $00 c, $00 c, $7E c, $00 c,  

$00 c, $7E c, $00 c, $00 c, $DF c, $00 c, $00 c, $DF c, $00 c,  

$01 c, $DF c, $80 c, $01 c, $FF c, $80 c, $03 c, $FF c, $C0 c,  

$07 c, $FF c, $E0 c, $07 c, $3C c, $E0 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 4b: Black bishop on black field 

$AA c, $AA c, $AA c, $55 c, $41 c, $55 c, $AA c, $98 c, $AA c,  

$55 c, $3C c, $55 c, $AA c, $7E c, $2A c, $54 c, $E7 c, $15 c,  

$A9 c, $C3 c, $AA c, $55 c, $C3 c, $95 c, $A8 c, $E7 c, $2A c,  

$54 c, $7E c, $55 c, $AA c, $3C c, $AA c, $55 c, $7E c, $55 c,  

$AA c, $7E c, $2A c, $54 c, $DF c, $55 c, $A8 c, $DF c, $2A c,  

$55 c, $DF c, $95 c, $A9 c, $FF c, $8A c, $53 c, $FF c, $C5 c,  

$A7 c, $FF c, $EA c, $57 c, $3C c, $E5 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $2A c, $55 c, $55 c, $55 c,  

\ 5a: White knight on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $33 c, $00 c, $00 c, $7F c, $80 c,  

$00 c, $C0 c, $C0 c, $01 c, $B0 c, $C0 c, $03 c, $30 c, $C0 c,  

$03 c, $00 c, $C0 c, $06 c, $00 c, $C0 c, $0C c, $F0 c, $C0 c,  

$0D c, $98 c, $C0 c, $07 c, $30 c, $E0 c, $00 c, $60 c, $E0 c,  

$00 c, $C1 c, $E0 c, $01 c, $81 c, $E0 c, $03 c, $01 c, $E0 c,  

$07 c, $FF c, $E0 c, $07 c, $00 c, $E0 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 5b: White knight on black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $00 c, $55 c, $AA c, $33 c, $2A c, $54 c, $7F c, $95 c,  

$A8 c, $C0 c, $CA c, $51 c, $B0 c, $D5 c, $A3 c, $30 c, $CA c,  

$53 c, $00 c, $D5 c, $A6 c, $00 c, $CA c, $4C c, $F0 c, $D5 c,  

$AD c, $98 c, $CA c, $47 c, $30 c, $E5 c, $A0 c, $60 c, $EA c,  

$54 c, $C1 c, $E5 c, $A9 c, $81 c, $EA c, $53 c, $01 c, $E5 c,  

$A7 c, $FF c, $EA c, $57 c, $00 c, $E5 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 6a: Black knight on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $00 c, $33 c, $00 c, $00 c, $7F c, $80 c,  

$00 c, $FF c, $C0 c, $01 c, $CF c, $C0 c, $03 c, $CF c, $C0 c,  

$03 c, $FF c, $C0 c, $07 c, $FF c, $C0 c, $0D c, $FF c, $C0 c,  

$0F c, $9F c, $C0 c, $07 c, $3F c, $E0 c, $00 c, $7F c, $E0 c,  

$00 c, $FF c, $E0 c, $01 c, $FF c, $E0 c, $03 c, $FF c, $E0 c,  

$07 c, $FF c, $E0 c, $07 c, $3C c, $E0 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 6b: White knight on black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$55 c, $00 c, $55 c, $AA c, $33 c, $2A c, $54 c, $7F c, $95 c,  

$A8 c, $FF c, $CA c, $51 c, $CF c, $D5 c, $A3 c, $CF c, $CA c,  

$53 c, $FF c, $D5 c, $A7 c, $FF c, $CA c, $4D c, $FF c, $D5 c,  

$AF c, $9F c, $CA c, $47 c, $3F c, $E5 c, $A0 c, $7F c, $EA c,  

$54 c, $FF c, $E5 c, $A9 c, $FF c, $EA c, $53 c, $FF c, $E5 c,  

$A7 c, $FF c, $EA c, $57 c, $3C c, $E5 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 7a: White rook on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $07 c, $BD c, $E0 c, $04 c, $A5 c, $20 c,  

$04 c, $E7 c, $20 c, $04 c, $00 c, $20 c, $07 c, $FF c, $E0 c,  

$01 c, $81 c, $80 c, $01 c, $81 c, $80 c, $01 c, $81 c, $80 c,  

$01 c, $81 c, $80 c, $01 c, $81 c, $80 c, $01 c, $81 c, $80 c,  

$01 c, $81 c, $80 c, $01 c, $FF c, $80 c, $03 c, $81 c, $C0 c,  

$07 c, $FF c, $E0 c, $07 c, $00 c, $E0 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 7a: White rook on black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$50 c, $00 c, $05 c, $A7 c, $BD c, $EA c, $54 c, $A5 c, $25 c,  

$A4 c, $E7 c, $2A c, $54 c, $00 c, $25 c, $A7 c, $FF c, $EA c,  

$51 c, $81 c, $95 c, $A9 c, $81 c, $AA c, $51 c, $81 c, $95 c,  

$A9 c, $81 c, $AA c, $51 c, $81 c, $95 c, $A9 c, $81 c, $AA c,  

$51 c, $81 c, $95 c, $A9 c, $FF c, $8A c, $53 c, $81 c, $C5 c,  

$A7 c, $FF c, $EA c, $57 c, $00 c, $E5 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 8a: Black rook on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

$00 c, $00 c, $00 c, $07 c, $BD c, $E0 c, $05 c, $AD c, $60 c,  

$07 c, $FF c, $E0 c, $07 c, $FF c, $E0 c, $07 c, $FF c, $E0 c,  

$01 c, $9F c, $80 c, $01 c, $9F c, $80 c, $01 c, $9F c, $80 c,  

$01 c, $9F c, $80 c, $01 c, $9F c, $80 c, $01 c, $9F c, $80 c,  

$01 c, $9F c, $80 c, $01 c, $FF c, $80 c, $03 c, $99 c, $C0 c,  

$07 c, $FF c, $E0 c, $07 c, $3C c, $E0 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 8a: Black rook on black field 

$AA c, $AA c, $AA c, $55 c, $55 c, $55 c, $AA c, $AA c, $AA c,  

$50 c, $00 c, $05 c, $A7 c, $BD c, $EA c, $55 c, $AD c, $65 c,  

$A7 c, $FF c, $EA c, $57 c, $FF c, $E5 c, $A7 c, $FF c, $EA c,  

$51 c, $9F c, $95 c, $A9 c, $9F c, $AA c, $51 c, $9F c, $95 c,  

$A9 c, $9F c, $AA c, $51 c, $9F c, $95 c, $A9 c, $9F c, $AA c,  

$51 c, $9F c, $95 c, $A9 c, $FF c, $8A c, $53 c, $99 c, $C5 c,  

$A7 c, $FF c, $EA c, $57 c, $3C c, $E5 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 9a: White queen on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $18 c, $00 c,  

$00 c, $24 c, $00 c, $01 c, $A5 c, $80 c, $02 c, $5A c, $40 c,  

$1A c, $5A c, $58 c, $25 c, $99 c, $A4 c, $25 c, $99 c, $A4 c,  

$19 c, $99 c, $98 c, $0C c, $DB c, $30 c, $0C c, $DB c, $30 c,  

$06 c, $DB c, $60 c, $06 c, $DB c, $60 c, $07 c, $FF c, $E0 c,  

$06 c, $00 c, $60 c, $07 c, $FF c, $E0 c, $06 c, $00 c, $60 c,  

$07 c, $FF c, $E0 c, $06 c, $00 c, $60 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 9b: White queen on black field 

$AA c, $AA c, $AA c, $55 c, $41 c, $55 c, $AA c, $98 c, $AA c,  

$54 c, $24 c, $15 c, $A9 c, $A5 c, $8A c, $42 c, $5A c, $41 c,  

$9A c, $5A c, $5A c, $25 c, $99 c, $A5 c, $A5 c, $99 c, $A4 c,  

$19 c, $99 c, $99 c, $8C c, $DB c, $32 c, $4C c, $DB c, $35 c,  

$A6 c, $DB c, $62 c, $56 c, $DB c, $65 c, $A7 c, $FF c, $EA c,  

$56 c, $00 c, $65 c, $A7 c, $FF c, $EA c, $56 c, $00 c, $65 c,  

$A7 c, $FF c, $EA c, $56 c, $00 c, $65 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 10a: Black queen on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $18 c, $00 c,  

$00 c, $2C c, $00 c, $01 c, $BD c, $80 c, $02 c, $DA c, $C0 c,  

$1B c, $DB c, $D8 c, $2D c, $99 c, $AC c, $3D c, $99 c, $BC c,  

$19 c, $99 c, $98 c, $0C c, $DB c, $30 c, $0C c, $DB c, $30 c,  

$06 c, $DB c, $60 c, $06 c, $DB c, $60 c, $07 c, $FF c, $E0 c,  

$06 c, $7E c, $60 c, $07 c, $FF c, $E0 c, $06 c, $7E c, $60 c,  

$07 c, $FF c, $E0 c, $06 c, $7E c, $60 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 10b: Black queen on black field 

$AA c, $AA c, $AA c, $55 c, $41 c, $55 c, $AA c, $98 c, $AA c,  

$54 c, $2C c, $15 c, $A9 c, $BD c, $8A c, $42 c, $DA c, $C1 c,  

$9B c, $DB c, $DA c, $2D c, $99 c, $AD c, $BD c, $99 c, $BC c,  

$19 c, $99 c, $99 c, $8C c, $DB c, $32 c, $4C c, $DB c, $35 c,  

$A6 c, $DB c, $62 c, $56 c, $DB c, $65 c, $A7 c, $FF c, $EA c,  

$56 c, $7E c, $65 c, $A7 c, $FF c, $EA c, $56 c, $7E c, $65 c,  

$A7 c, $FF c, $EA c, $56 c, $7E c, $65 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 11a: White king on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $18 c, $00 c,  

$00 c, $18 c, $00 c, $00 c, $66 c, $00 c, $00 c, $66 c, $00 c,  

$00 c, $18 c, $00 c, $03 c, $18 c, $C0 c, $07 c, $99 c, $E0 c,  

$0C c, $FF c, $30 c, $0C c, $C3 c, $30 c, $0C c, $66 c, $30 c,  

$06 c, $3C c, $60 c, $03 c, $00 c, $C0 c, $07 c, $FF c, $E0 c,  

$06 c, $00 c, $60 c, $07 c, $FF c, $E0 c, $06 c, $00 c, $60 c,  

$07 c, $FF c, $E0 c, $06 c, $00 c, $60 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 11b: White King on black field 

$AA c, $AA c, $AA c, $55 c, $41 c, $55 c, $AA c, $98 c, $AA c,  

$55 c, $18 c, $55 c, $AA c, $66 c, $2A c, $54 c, $66 c, $15 c,  

$A8 c, $18 c, $0A c, $53 c, $18 c, $C5 c, $A7 c, $99 c, $E2 c,  

$4C c, $FF c, $35 c, $AC c, $C3 c, $32 c, $4C c, $66 c, $35 c,  

$A6 c, $3C c, $62 c, $53 c, $00 c, $C5 c, $A7 c, $FF c, $EA c,  

$46 c, $00 c, $65 c, $A7 c, $FF c, $EA c, $56 c, $00 c, $65 c,  

$A7 c, $FF c, $EA c, $56 c, $00 c, $65 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

\ 12a: Black king on white field 

$00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c, $18 c, $00 c,  

$00 c, $18 c, $00 c, $00 c, $7E c, $00 c, $00 c, $7E c, $00 c,  

$00 c, $18 c, $00 c, $03 c, $18 c, $C0 c, $07 c, $99 c, $E0 c,  

$0F c, $FF c, $F0 c, $0C c, $DB c, $30 c, $0F c, $E7 c, $F0 c,  

$07 c, $FF c, $E0 c, $03 c, $FF c, $C0 c, $07 c, $FF c, $E0 c,  

$06 c, $7E c, $60 c, $07 c, $FF c, $E0 c, $06 c, $7E c, $60 c,  

$07 c, $FF c, $E0 c, $06 c, $7E c, $60 c, $07 c, $FF c, $E0 c,  

$03 c, $FF c, $C0 c, $00 c, $00 c, $00 c, $00 c, $00 c, $00 c,  

\ 12b: Black King on black field 

$AA c, $AA c, $AA c, $55 c, $41 c, $55 c, $AA c, $98 c, $AA c,  

$55 c, $18 c, $55 c, $AA c, $66 c, $2A c, $54 c, $66 c, $15 c,  

$A8 c, $18 c, $0A c, $53 c, $18 c, $C5 c, $A7 c, $99 c, $E2 c,  

$4F c, $FF c, $F5 c, $AC c, $DB c, $32 c, $4F c, $E7 c, $F5 c,  

$A7 c, $FF c, $E2 c, $53 c, $FF c, $C5 c, $A7 c, $FF c, $EA c,  

$46 c, $7E c, $65 c, $A7 c, $FF c, $EA c, $56 c, $7E c, $65 c,  

$A7 c, $FF c, $EA c, $56 c, $7E c, $65 c, $A7 c, $FF c, $EA c,  

$53 c, $FF c, $C5 c, $A8 c, $00 c, $0A c, $55 c, $55 c, $55 c,  

 

\ ------------------------------------------------------------- 

\ Fields and variables 

\ ------------------------------------------------------------- 

\ Save chessmoves in an array 

200 constant #maxmoves \ actual maximum 200 moves 

variable chessmoves    \ actual move position 

variable maxmoves      \ maximum moves logged 

create movelist ( old pos; figure; new pos; replaced figur ) 

   #maxmoves 4 * allot 

variable fromfield \ flag for moving: -1: new; 0-63: first position 

\ Chess start board 

create startboard 

   7 c,  5 c,  3 c,  9 c, 11 c,  3 c,  5 c,  7 c, 

   1 c,  1 c,  1 c,  1 c,  1 c,  1 c,  1 c,  1 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   2 c,  2 c,  2 c,  2 c,  2 c,  2 c,  2 c,  2 c, 

   8 c,  6 c,  4 c, 10 c, 12 c,  4 c,  6 c,  8 c, 

 

\ Actual displayed board 

create dispboard 

   7 c,  5 c,  3 c,  9 c, 11 c,  3 c,  5 c,  7 c, 

   1 c,  1 c,  1 c,  1 c,  1 c,  1 c,  1 c,  1 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c,  0 c, 

   2 c,  2 c,  2 c,  2 c,  2 c,  2 c,  2 c,  2 c, 

   8 c,  6 c,  4 c, 10 c, 12 c,  4 c,  6 c,  8 c, 

 

\ ------------------------------------------------------------- 

\ Draw board (0 = a1 left bottom is black) 

\ ------------------------------------------------------------- 

: drawfield ( field figure -- ) \ draw one field 

  2* >r  dup dup 3 rshift + 1 and  0= if r> 1 +  >r then \ Figure+color 

  63 over -  3 rshift  768 * 

  swap       7 and       3 *  +  lcd-vram + \ LCD address 

  r> 72 *  icons  +                         \ icon Address 

  24 0 do 

    3 0 do dup >r c@  over c!  1 +  r> 1 + loop 

    >r 29 +  r> 

  loop 

  drop drop ; 

 

: pos.      ( field -- ) \ print a1..h8 

  dup 7 and  $61 + lcd-emit 

  3 rshift   $31 + lcd-emit ; 

 

: drawboard ( -- ) 

  lcd-cls 

  64 0 do i  i dispboard + c@  drawfield loop 

  fieldsize 8 *  0              lcd-move-to 

  fieldsize 8 *  dup            lcd-line-to 

  0              fieldsize 8 *  lcd-line-to 

  25  0   lcd-at-xy   s" Chess"                       lcd-type 

  25  1   lcd-at-xy   chessmoves @ 0 <# # # # # # #>  lcd-type 

  25  2   chessmoves @  dup  dup 13 min  -     \ last 10 moves 

  ?do over over lcd-at-xy  1 + 

      i 4 * movelist +  dup c@ pos.  s" -" lcd-type  2 + c@ pos. 

  loop drop drop 

  9   lcd-text-rows 1-  lcd-at-xy   s" Back  Again  Exit" lcd-type ; 

 

\ ------------------------------------------------------------- 

\ Main program with move handling 

\ ------------------------------------------------------------- 

: allowfrom?   ( field -- field -1 | 0 ) \ check from 

  dup dispboard + c@          \ field figure 

  ?dup 0= if drop 0 exit then \ no empty field 

  chessmoves @  xor  1 and    \ right color ? 

  if -1 else drop 0  then ;  

 

: allowto?     ( field -- field -1 | 0 ) \ check to 

  fromfield @  over  = if drop 0 exit then    \ same field not allowed 

  dup dispboard + c@  ?dup 0= if -1 exit then \ empty field allowed 

  chessmoves @  xor  1 and                    \ right color ? 

  if drop 0 else -1 then ;     

 

: -move   ( -- ) \ backward one move 

  chessmoves @  ?dup 

  if   1 -  dup chessmoves !  4 *  movelist +   \ movelist 

           dup 1 + c@  over c@  dispboard +  c! \ back to old position 

       2 + dup 1 + c@  swap c@  dispboard +  c! \ restore field 

       drawboard 

  then ; 

: +move   ( -- ) \ forward one move 

  chessmoves @   dup  maxmoves @  < 

  if   dup  1 +  chessmoves !  4 *  movelist +  \ movelist 

       0           over     c@  dispboard +  c! \ remove from old position 

       dup 1 + c@  swap 2 + c@  dispboard +  c! \ set to new field 

       drawboard 

  else drop 

  then ; 

 

: setmove ( from to -- ) \ save move 

  chessmoves @   dup  #maxmoves  < 

  if   dup  1 +  dup chessmoves !  maxmoves ! 

           4 * movelist +    >r 

       over                           r@     c! \ old position 

       swap dispboard +  dup c@   dup r@ 1 + c! \ figure 

       >r  0 swap c!  r>                        \ empty old positon 

       over                           r@ 2 + c! \ new position 

       swap dispboard +  dup c@       r> 3 + c! \ replaced figure 

                  c!                            \ figure to new position 

       drawboard 

  else drop drop drop 

  then ; 

 

: newmove   ( -- ) \ with highlight of first field 

  ctp-pos ctp-flush 

  dup fieldsize 8 * 0 within  over fieldsize 8 * 0 within or 

  if drop drop exit then 

  fieldsize /  7 swap -  8 *  swap fieldsize /  + 

  fromfield @  -1  = 

  if   allowfrom? 0= if exit then   \ ignore wrong figure/field 

       dup fromfield !                 \ save position 

       dup 7 and                 fieldsize *  \ x*size 

       swap  3 rshift  7 swap -  fieldsize *  \ y*size 

       over over  lcd-move-to 

       over                 over fieldsize + 1-  lcd-line-to 

       over fieldsize + 1-  over fieldsize + 1-  lcd-line-to 

       over fieldsize + 1-  over                 lcd-line-to 

                                                 lcd-line-to 

  else allowto? if fromfield @  swap  setmove   -1 fromfield ! then 

  then ; 

 

: chess ( -- ) 

  startboard dispboard 64 cmove 

  0 chessmoves !  0 maxmoves ! 

  drawboard 

  button-flush   ctp-flush   -1 fromfield ! 

  begin 

     wait-for-event 

     button? if button  button-flush  -1 fromfield ! 

                dup button-right  = if drop exit then 

                dup button-left   = if -move then 

                    button-centre = if +move then 

             then 

     ctp-pos? if newmove then 

  again ; 

 

\ ------------------------------------------------------------- 

 

base ! 

 


